# FindSequenceFunction for sum of hypergeometric terms?

The built-in function FindSequenceFunction is quite good at recognizing hypergeometric terms, i.e. terms $c_k$ for which $c_{k+1}/c_k$ is a rational function of $k$.

FindSequenceFunction[CoefficientList[Series[
Hypergeometric2F1[1/3(2-I Sqrt[2]), 1/3(2 + I Sqrt[2]), 1/3, x],
{x, 0, 10}] // Normal, x] // Simplify]


evaluates nicely to:

(*
(Pochhammer[2/3 - (I Sqrt[2])/3, -1 + #1] Pochhammer[2/3 + (I Sqrt[2])/3, -1 + #1]) /
(Pochhammer[1/3, -1 + #1] Pochhammer[1, -1 + #1]) &
*)


Nonetheless, I haven't been able to get Mathematica to recognize the sum of two hypergeometric terms. For example:

FindSequenceFunction[CoefficientList[Series[
Hypergeometric2F1[1/3 (2 - I Sqrt[2]), 1/3 (2 + I Sqrt[2]), 1/3, x] +
Hypergeometric2F1[1/3 (4 - I Sqrt[2]), 1/3 (4 + I Sqrt[2]), 5/3, x],
{x, 0, 10}] // Normal, x] // Simplify]


returns:

(*
FindSequenceFunction[{2, 16/5, 141/40, 5653/1540, 46239/12320, 217707/57200, 11190509/2912000, 17236629861/4453904000, 277267543443/71262464000, 2883838646953/738075520000, 22686466063740149/5786512076800000}]
*)


I appreciate that identifying sums of hypergeometric terms might be more difficult than recognizing one such term because the ratio of successive terms is then no longer a rational function of the index.

Question: Is it possible to trick Mathematica into trying a little (or a lot) harder to recognize the sum of two hypergeometric terms?

It might be better just to ask SeriesCoefficient for the answer. From what I can tell it should always find the answer for functions of hypergeometric type.

expr = Hypergeometric2F1[1/3 (2 - I Sqrt[2]), 1/3 (2 + I Sqrt[2]), 1/3, x] +
Hypergeometric2F1[1/3 (4 - I Sqrt[2]), 1/3 (4 + I Sqrt[2]), 5/3, x];

Expand[Refine[SeriesCoefficient[expr, {x, 0, n}], n >= 0]] /. {
Gamma[c_ + n] :> Gamma[c] Pochhammer[c, n]
}

(Pochhammer[2/3 - (I Sqrt[2])/3, n] Pochhammer[2/3 + (I Sqrt[2])/3,n]) /
(Pochhammer[1/3, n] Pochhammer[1, n]) +
(Pochhammer[4/3 - (I Sqrt[2])/3, n] Pochhammer[4/3 + (I Sqrt[2])/3, n]) /
(Pochhammer[1, n] Pochhammer[5/3, n])


Let me point out my paper http://arxiv.org/pdf/1301.6617.pdf (also http://iopscience.iop.org/article/10.1088/1751-8113/46/44/445302/pdf). Figure 3--containing SIX hypergeometric functions--in it is the result of an application of FindSequenceFunction, as explained in the paper. The displayed output was obtained using various transformation rules from the original FindSequenceFunction output--also containing multiple hypergeometric terms. (I would have posted this as a comment, but lacked sufficient reputation points to do so.)